



Videoton TV COMPUTER

Orgona3

Háromcsatornás hanggenerátor működése

©2023 by Szalontai Béla (Fast and Force software)

Miért 3 csatorna, miért nem 2 vagy 4?

Két oka van.


- Az egyik, hogy minimalizálni kell a megszakítás teljes végrehajtási ciklus idejét annak érdekében, hogy minél sűrűbben legyen hívható a megszakítás, ezáltal pontosabb legyen a hang generálás és magasabb frekvenciák is kiadhatók legyenek. Ezért a hanggeneráláshoz szükséges csatornánként kettő bájtot (frekvenciaosztó és pillanatnyi hangerő) az EX regiszterekben tárolom. Így az árnyék HL, DE, BC regiszterek felhasználhatók, azaz összesen 6 db egy bájtos regiszterem van. Ez pont elég 3 csatornához. 4 csatorna esetén már valahol máshol kellene tárolni az aktuális frekvenciaosztót és a pillanatnyi hangerőt, ami sokkal költségesebb művelet. Regiszterekkel dolgozni sokkal gyorsabb, mint memóriacímeken lévő adatokkal. Természetesen megoldható a 4 csatorna is, de annak az ára, hogy megnő a megszakítás teljes végrehajtási ciklusideje, és emiatt lejjebb kell venni a megszakítás frekvenciáját. Ezzel csökken a hang előállítási pontosság és az alsó tartományban a hangok már túl mélyek, torzak lesznek.
- A másik ok, hogy a TVC 4 bites DA konverterébe a csatornákat mixelni kell. **Hang csatornák mixeléséhez a legjobb módszer az egyes csatornák hangerő értékeinek egyszerű összeadása.** Valójában ennél is pontosabb lenne, ha összeadnánk a csatornák hangerő értékeit, majd osztanánk a csatornák számával. De az osztás, még 2-vel osztható számokkal is túl költséges művelet, 3 csatornánál pedig csak táblázattal lehet 3-mal osztani. Így maradt az egyszerű összeadás. Mivel a DA konverter 4 bitje az adott porton az 5. 4. 3. és 2. bit, így a mixelt hangerőt még kettővel balra is kellene shift-elni, ami szintén „költséges”. **Az én módszerem az, hogy decimális 20 a maximális hangerő csatornánként, ami 3 csatorna esetén összeadva pont 60-at fog eredményezni. TEHÁT ha a három csatorna pillanatnyi hangerejét összeadom, akkor sem osztani nem kell hárommal, sem fel shift-elni a DA konverter megfelelő bitjeire, hanem egyből a helyes eredményt kapom, azaz megspóroltam egy rakás órajel ciklust és a csatornák összeadásával tökéletes mixelést tudok megvalósítani.** Ha 0..15-ig menne a hangerő csatornánként, akkor az összeadás után még balra kéne shift-telni kétszer, hogy a megfelelő bithelyeken legyen a DA konverterre kiadandó adat.

A mixer kódja összesen így 37 utasítás ciklus

MIXER	LD	A,D	; 4	
	ADD	A,B	; 4	
	ADD	A,C	; 4	
	AND	60	; 7	mask
	OR	1	; 7	must have in 4 color graphics mode
	OUT	(6),A	; 11	write to sound port

Miért négyszögjel?

- A leggyorsabban (legkevesebb utasítással és ciklusidővel) előállítható hullámforma a négyszögjel, mivel annak egy teljes hullámát 2 ciklus alatt elő lehet állítani, így ahhoz, hogy például egy normál zenei 440Hz-es A hangot kiadjunk az alábbi frekvencia osztóval kell dolgoznunk: $IRQ \text{ freq} / 440 \text{ Hz} / 2 = 13950 / 440 / 2 = 15,85$ ami kerekítve **16**. Ennél a hangnál a hiba már kb. 8Hz de ez még emberi fül által alig hallható.



1.ciklus 2. ciklus

- Hogyan lesz ebből hang? Másodpercenként 13950-szer lefut a megszakítás. A megszakításban, ha kiadunk egy 1-et a D/A konverterre, várunk 16 db megszakítás ciklust, majd kiadunk 0-át a D/A konverterre, majd újra várunk 16 db ciklust és így tovább, akkor abból egy $3125000 / 13950 * 2 = 448$ Hz-es négyszög hullámformájú hang fog megszólalni.

Hogyan működik a hanggenerátor megszakítás?

A megszakítás kódja dinamikusan változik. Erre azért van szükség, hogy ne kelljen változókat vagy regisztereket használni bizonyos funkciók megvalósításához.

Ennek a megértéséhez felvázolok egy példát: normál esetben, ha egy csatornát le szeretnénk tiltani, akkor definiálunk egy CH_ENABLE változót, és a változó értékének megfelelően a megszakítás kódja dönti el, hogy a csatorna szóljon vagy sem. Ehhez például azonban ezekre a költséges utasításokra lenne szükség:

```
LD    A, (CH_ENABLE)
OR    A
JR    Z, NEXT_CHANNEL
```

ez $13+4+12 = 29$ utasítás ciklus

míg, ha csak egyetlen JR nn utasítást használunk, akkor az **csak 12 ciklus!** nn helyére persze nem mindegy mit írunk dinamikusan. Ha \$00-t, akkor folytatja a következő utasításon, azaz előre ugrik 1-et, ha pedig a következő csatornán szeretnénk folytatni a végrehajtást, akkor oda kell ugrani, relatív címmel. A mi esetünkben 15 bájtnyi utasítást kell átugranunk, így a csatorna tiltása vagy engedélyezése a következő kódból áll:

```
DB    $18          ; ez egy JR utasítás
DB    $0F          ; JR $0F vagy JR $00, attól függ, hogy ← ide mit írunk bele
```

A dinamikusan változó kód az alábbi, a megszakítás rutin kódjában definiált változókkal működtethető:

CH1_ENABLED -> ha ide \$00-t írunk, akkor a csatorna engedélyezve van, ha \$0F-et, akkor tiltva

CH1_DIVIDER -> ide kell írni az adott csatorna frekvencia osztó értékét

CH1_VOLUME -> ez pedig a csatorna hangereje. helyett (0..20-ig)

CH2_ENABLED

CH2_DIVIDER

CH2_VOLUME

CH3_ENABLED

CH3_DIVIDER

CH3_VOLUME

A megszakítás frekvenciáját úgy kell belőni, hogy a legrosszabb esetben se (worst case cycle time) fusson a megszakítás önmagára, sőt leginkább jó lenne, ha maradna még processzor idő az egyéb műveletekre is, nem csak a hang generálására. A jelenlegi összeállításban a megszakítás frekvenciája **13950 Hz**. Így a legrosszabb esetben is „csak” 96%-át viszi el a processzor időnek a hanggenerálás (azaz a megszakítás lefutása). De ez egy dinamikusan változó százalék lásd később.

Megvan tehát a megszakítás frekvenciája, most már kiszámolhatók az adott zenei hanghoz tartozó frekvencia osztók. Ehhez a számításokat a mellékelt [calculations.xls](#) excel táblában találhatjuk.

Például a zongora 2. oktávján megszólaló **Fisz** hanghoz tartozó frekvencia osztót az alábbi módon tudjuk kiszámolni:

F#2 -> 92,5Hz -> $3125000 \text{ Hz} / 92,5 \text{ Hz} / 2 = 75$

(a 2-es osztó azért kell, mert 2 félhullámból áll egy négyszögjel azaz 1-ből és 0-ból) A **F#2** hang frekvencia osztója tehát **75**, ha ezt betesszük a **CH1_DIVIDER** változóba (a megszakítás rutin dinamikus kódja), akkor az 1-es csatornán egy zenei **F#2** hang fog megszólalni. Persze ehhez a hangerőt (**CH1_VOLUME**) is be kell állítani 0-nál nagyobb értékre és a csatornát is engedélyezni kell: **CH1_ENABLED** = \$00

A megszakítás frekvenciája minél magasabb, annál jobb, lásd a fentieket. Esetünkben ez **13950 Hz** is lehet, hiszen a megszakítás lefutásának teljes ciklusideje maximum **215**.

Ennek azonban van egy nagyon fontos feltétele: minden csatornán más-más frekvenciájú hangnak kell szólania.

Miért? Ha megnézzük a megszakítás rutint vannak olyan részei, amelyek mindig lefutnak és vannak olyan részek, amelyek csak bizonyos feltételek teljesülésénél futnak le.

sárgával azok a részek kiemelve, amelyek mindig lefutnak (**121** ciklus):

1A06	08	IT_HANDLER	EX AF,AF'	[4]	
1A07	D9		EXX	[4]	
1A08	D307		OUT (7),A	[11]	
1A0A	18		DB \$18	[12]	; JR nn utasítás
1A0B	0F	CH1_ENABLED	DB \$0F		
1A0C	1D		DEC E	[4]	
1A0D	200C		JR NZ,MIH1	[12]	
1A0F	1E		DB \$1E		
1A10	01	CH1_DIVIDER	DB \$01		
1A11	7A		LD A,D	[4]	
1A12	B7		OR A	[4]	
1A13	2004		JR NZ,MIH0_1	[12]	
1A15	16		DB \$16		
1A16	00	CH1_VOLUME	DB \$00		
1A17	1802		JR MIH1	[12]	
1A19	1600	MIH0_1	LD D,0	[7]	
1A1B	18	MIH1	DB \$18	[12]	; JR nn utasítás
1A1C	0F	CH2_ENABLED	DB \$0F		
1A1D	25		DEC H	[4]	
1A1E	200C		JR NZ,MIH2	[12]	
1A20	26		DB \$26		
1A21	02	CH2_DIVIDER	DB \$02		
1A22	78		LD A,B	[4]	
1A23	B7		OR A	[4]	
1A24	2004		JR NZ,MIH1_1	[12]	
1A26	06		DB \$06		
1A27	00	CH2_VOLUME	DB \$00		
1A28	1802		JR MIH2	[12]	
1A2A	0600	MIH1_1	LD B,0	[7]	
1A2C	18	MIH2	DB \$18	[12]	; JR nn utasítás
1A2D	0F	CH3_ENABLED	DB \$0F		
1A2E	2D		DEC L	[4]	
1A2F	200C		JR NZ,MIXER	[12]	
1A31	2E		DB \$2E		
1A32	03	CH3_DIVIDER	DB \$03		
1A33	79		LD A,C	[4]	
1A34	B7		OR A	[4]	
1A35	2004		JR NZ,MIH2_1	[12]	
1A37	0E		DB \$0E		
1A38	00	CH3_VOLUME	DB \$00		

1A39	1802		JR	MIXER	[12]
1A3B	0E00	MIH2_1	LD	C,0	[7]
1A3D	7A	MIXER	LD	A,D	[4]
1A3E	80		ADD	A,B	[4]
1A3F	81		ADD	A,C	[4]
1A40	E63C		AND	60	[7]
1A42	F601		OR	1	[7]
1A44	D306		OUT	(6),A	[11]
1A46	D9		EXX		[4]
1A47	08		EX	AF,AF'	[4]
1A48	FB		EI		[4]
1A49	C9		RET		[10]

Ha mindhárom csatornán más-más frekvenciájú hang szól, az azt jelenti, hogy az osztók számlálói más-más időpillanatban nullázódnak, ezért egyszerre csak egy csatornát kell lekezelni, már ami a legköltségesebb műveletet illet, azaz a négyszögjel váltását. Tehát amíg az adott csatorna frekvencia számlálója számol, azaz csökken egyel, de még nem érte el a nullát, addig azon a csatornán nincs teendő, az éppen aktuális 1 vagy 0 azaz a beállított hangerő vagy 0 hangerő fog kijutni a mixerre. Amint a frekvencia számláló leszámolt 0-ra, úgy 1-ről nullára vagy 0-ról egyre billentjük a négyszögjel alakját.

Ha egy csatornát engedélyezünk, akkor már a **zölddel** jelölt utasítások fognak lefutni, persze nem mindegyik, de legrosszabb esetben ez maximum **215** ciklus.

Ha mindhárom csatorna engedélyezve van és biztosítjuk, hogy minden csatornán más-más időpontban járjon le a számláló, és mindhárom számláló még számol (nagyobb mint 1), akkor a megszakítás teljes ciklus ideje: **169**

Ha mindhárom csatorna engedélyezve van és biztosítjuk, hogy minden csatornán más-más időpontban járjon le a számláló, akkor a teljes megszakítás ciklusa, ha egy számláló éppen lejárt: **215**

Tehát a megszakítás teljes végrehajtási ideje 121, 169, 215 ciklus között váltakozik. Az idő nagy részében, ha mindhárom hang szól, akkor 169, amikor egy-egy frekvencia számláló leszámolt akkor pedig 215. Ha egyetlen hang sem szól, akkor 121.

Hogyan működik a főprogram?

A főprogram felelős a billentyűzet mátrix figyeléséért, a hangok be és kikapcsolásáért és a hangerő változásáért.

- A hangerő lecsengéséhez 20-tól 0-ig folyamatosan csökkenő számot kell írnia a CHn_VOLUME változók valamelyikébe. A lecsengéshez a számlálók az **AD1**, **AD2** és **AD3** változók.
- A Vibrato előállításához a CHn_DIVIDER változót növeljük majd csökkentjük felváltva egyel. Ehhez a **CH1_FREQ_MOD**, **CH2_FREQ_MOD**, **CH3_FREQ_MOD** segédváltozókat használja a főprogram.

A Billentyűzet figyelése után, ha érzékelünk egy-egy lenyomott billentyűt, akkor egy KEY_HANDLER nevű szubrutin kerül meghívásra, ami gondoskodik a csatornák kezeléséről az alábbi algoritmus szerint:

- Ha egy hang már szól valamelyik csatornán, akkor az adott csatornán újra indítjuk az adott hangot, vagyis, ha a billentyűt nyomva tartjuk, akkor nincs hangerő lecsengés, hanem tartjuk (Sustain) a hangot.
- Ha az adott hang még nem szól, akkor megnézzük, hogy van-e szabad csatorna, amin éppen semmilyen hang nem szól. Ha van ilyen, akkor ott kiadjuk az adott hangot
- Ha egyik előző feltétel sem teljesül, akkor megnézzük, hogy melyik csatornán szól az adott hang a RELEASE_VOL_CONST -nál halkabban, és azt a csatornát újból befoglaljuk és az új frekvenciát adjuk ki rajta.

Ez az algoritmus biztosítja, hogy bármilyen billentyűt nyomunk le az ORGONÁN, mindig meg fog szólalni, még azon az áron is, ha az előzőleg megszólaló hang helyén fog szólalni.

A tényleges hang kiadását, azaz a megszakítás dinamikus változtatását a **KEY_PLAY_CH1** **Ch2** és **Ch3** kódsorok végzik.

Grafika

A 4 színű üzemmódot, a paletta és a keretszín beállítását a **GRAPHICS_CONFIG** rutin végzi.

Az egyszerűség kedvéért minden megjelenő kép ZX7 tömörített formában van letárolva, lásd a forráskód végén található adatokat (**INTRO_PICTURE_DATA**, **MAIN_PICTURE_DATA**, **HELP_PICTURE_DATA**)

A képernyő törlés grafikus effektért a **CLRSCR** szubrutin felel. Ezt még 1988-ban írtam, és ha jól emlékszem a Treasure Hunting játékban használtam először.

A képeket az **INTRO_PICTURE**, a **MAIN_PICTURE** és a **HELP_PICTURE** szubrutinok pakolják ki, ezek gondoskodnak a video memória belapozásáért is.

A ZX7 adatok kicsomagolásáért a **ZX7_DECOMPRESS** nevű szubrutin felel.

A SPACE és a CTRL gombok megnyomásakor a képen található szöveget simán XOR utasítással negálom, így vizualizálom a gombok megnyomását. Ezekért a **ZENGETO_XOR** és a **VIBRATO_XOR** szubrutinok felelnek.

Vége

A program végén (ESC megnyomásakor hívódik a **VEGE** címke). Itt visszaállítjuk az eredeti megszakítás frekvenciát és rutint, kiadunk egy Graphics 4 parancsot, visszaállítjuk a keret színét és a mentett árnyék regisztereket is.

Kérdés esetén keressetek a FaceBook-on!

2023.12.22